



SAMPLE PRINTER OUTPUT

TEC PRINTER ROM

FILE: print_3.rom

PROJECT DESIGNED AND WRITTEN BY KEN STONE
CODE ANNOTATION BY BRIAN CHIHA

THE PRINTER/PLOTTER ADD-ON FOR THE TEC WAS PUBLISHED IN ISSUE 12 OF THE TALKING ELECTRONICS MAGAZINE. IT CONNECTS THE TEC TO AN 8-BIT PRINTER/PLOTTER. AN EXPANSION BOARD THAT INCLUDED A ROM WHICH HAS ROUTINES TO ASSIST IN INTERFACING WITH THE PRINTER. THIS LISTING IS THAT ROM. THE ROM IS TO BE PLACED AT MEMORY LOCATION 1800H.

THE ROM HAS BEEN UPGRADED THREE TIMES, EACH TO ACCOMODATE A DIFFERENT MONITOR. THE FIRST PART IS FOR MON1/1A/1B, THE MON2 AND THE THIRD FOR JMON. MOST OF THE CHANGES CENTER AROUND THE DIFFERENT WAY EACH MONITOR HANDLES KEYBOARD INPUT

TO TALK TO A PRINTER IS SIMPLE. JUST SEND EITHER AN INSTRUCTION OR ASCII 8 BIT HEX VALUE TO THE PRINTERS PARALLEL PORT. FOR INSTANCE, A HEX VALUE OF '0D' TELLS THE PRINTER TO DO A CARRIAGE RETURN. WHILE A HEX VALUE OF '42', WILL PRINT THE LETTER 'B'. FOR LINE GRAPHICS, ONCE GRAPHICS MODE HAS BEEN SET, X,Y LINE COORDINATES ARE SENT AS ASCII TO THE PRINTER, IE: 38 30 2C 34 30 SENDS 80,40.

MON 1/1A/1B PRINTER CODE

IF USING MON 1/1A/1B, FOUR ROUTINES CAN BE CALLED:

| PROGRAM | ADDR | DESCRIPTION |
|------------------|----------|--|
| ----- | ---- | ----- |
| 1. HEX DUMP | - 1800 - | GIVEN A START ADDRESS, OUTPUT THE DATA TO THE PRINTER. THIS ROUTINE WILL OUTPUT THE ADDRESS AND 8 BYTES PER LINE. |
| 3. AUTO PRINT | - 1880 - | OUTPUT BYTES FROM ADDRESS 0800 UNTIL BYTE 'FF' IS REACHED |
| 3. KEN'S ROUTINE | - 18A0 - | SAMPLE PRINTER OUTPUT WRITTEN BY KEY STONE |
| 2. MANUAL PRINT | - 1980 - | TYPE IN ASCII OR CONTROL CODES DIRECTLY TO THE PRINTER |

START OF HEX DUMP ROUTINE. USER KEYS IN THE START ADDRESS AND IT IS STORED IN 'DE'
THEN THE ADDRESS AND 8 BYTES ARE PRINTER PER LINE. EG:

-> 1A50 3E 0D D3 06 2A 98 08 7C
-> 1A58 CD 82 1A 7D CD 82 1A 06

| | | |
|---------------|------------|---|
| 1800 3E 0D | LD A,0D | ;LOAD REGISTER A WITH CARRIAGE RETURN |
| 1802 D3 06 | OUT (06),A | ;SEND CR TO PRINTER |
| 1804 3E 0A | LD A,0A | ;LOAD REGISTER A WITH LINE FEED |
| 1806 D3 06 | OUT (06),A | ;SEND LF TO PRINTER |
| 1808 76 | HALT | ;HALT THE CPU AND WAIT FOR KEYBOARD INPUT (NMI) |
| 1809 ED 57 | LD A,I | ;LOAD KEY INPUT INTO REGISTER A |
| 180B 17 | RLA | ;ROTATE LEFT 4 TIMES TO MOVE |
| 180C 17 | RLA | ;THE LOW NIBBLE TO THE HIGH |
| 180D 17 | RLA | ;NIBBLE AND VICE VERSA |
| 180E 17 | RLA | ;IE: '04' BECOMES '40' |
| 180F 57 | LD D,A | ;SAVE KEY IN D |
| 1810 CD 5D 18 | CALL 185D | ;CALL CONVERT HEX TO ASCII CHARACTER AND PRINT ROUTINE ;(HIGH) |
| 1813 76 | HALT | ;HALT THE CPU AND WAIT FOR KEYBOARD INPUT (NMI) |
| 1814 ED 57 | LD A,I | ;LOAD KEY INPUT INTO REGISTER A |
| 1816 82 | ADD A,D | ;ADD FIRST KEY TO SECOND KEY |
| 1817 57 | LD D,A | ;STORE FIRST PART OF ADDRESS IN D |
| 1818 CD 61 18 | CALL 1861 | ;CALL CONVERT HEX TO ASCII CHARACTER AND PRINT ROUTINE ;(LOW) |
| 181B 76 | HALT | ;HALT THE CPU AND WAIT FOR KEYBOARD INPUT (NMI) |
| 181C ED 57 | LD A,I | ;LOAD KEY INPUT INTO REGISTER A |
| 181E 17 | RLA | ;ROTATE LEFT 4 TIMES TO MOVE |
| 181F 17 | RLA | ;THE LOW NIBBLE TO THE HIGH |
| 1820 17 | RLA | ;NIBBLE AND VICE VERSA |
| 1821 17 | RLA | ;IE: '04' BECOMES '40' |
| 1822 5F | LD E,A | ;SAVE KEY IN E |
| 1823 CD 5D 18 | CALL 185D | ;CALL CONVERT HEX TO ASCII CHARACTER AND PRINT ROUTINE ;(HIGH) |
| 1826 76 | HALT | ;HALT THE CPU AND WAIT FOR KEYBOARD INPUT (NMI) |
| 1827 ED 57 | LD A,I | ;LOAD KEY INPUT INTO REGISTER A |
| 1829 83 | ADD A,E | ;ADD THRID KEY TO FOURTH KEY |
| 182A 5F | LD E,A | ;STORE SECOND PART OF ADDRESS IN E |
| 182B CD 61 18 | CALL 1861 | ;CALL CONVERT HEX TO ASCII CHARACTER AND PRINT ROUTINE ;(LOW) |
| 182E C3 49 18 | JP 1849 | ;JUMP TO BYTE DUMP ROUTINE |

OUTPUT CURRENT ADDRESS 'DE' TO THE PRINTER

| | | |
|------------|------------|---------------------------------------|
| 1831 3E 0D | LD A,0D | ;LOAD REGISTER A WITH CARRIAGE RETURN |
| 1833 D3 06 | OUT (06),A | ;SEND CR TO PRINTER |

```

1835 3E 0A      LD A,0A      ;LOAD REGISTER A WITH LINE FEED
1837 D3 06      OUT (06),A    ;SEND LF TO PRINTER
1839 7A         LD A,D        ;LOAD MSB ADDRESS IN REGISTER A
183A CD 5D 18    CALL 185D     ;CALL CONVERT HEX TO ASCII CHARACTER AND PRINT ROUTINE
                                ;(HIGH)
183D 7A         LD A,D        ;LOAD MSB ADDRESS IN REGISTER A
183E CD 61 18    CALL 1861     ;CALL CONVERT HEX TO ASCII CHARACTER AND PRINT ROUTINE
                                ;(LOW)
1841 7B         LD A,E        ;LOAD LSB ADDRESS IN REGISTER A
1842 CD 5D 18    CALL 185D     ;CALL CONVERT HEX TO ASCII CHARACTER AND PRINT ROUTINE
                                ;(HIGH)
1845 7B         LD A,E        ;LOAD LSB ADDRESS IN REGISTER A
1846 CD 61 18    CALL 1861     ;CALL CONVERT HEX TO ASCII CHARACTER AND PRINT ROUTINE
                                ;(LOW)

```

HEX DUMP 8 BYTES TO THE PRINTER

```

1849 06 08      LD B,08      ;LOAD B WITH THE NUMBER OF DATA BYTES TO PRINT
184B 3E 20      LD A,20      ;LOAD REGISTER A WITH AN ASCII SPACE " "
184D D3 06      OUT (06),A    ;SEND A SPACE CHARACTER TO THE PRINTER
184F 1A         LD A,(DE)     ;LOAD REGISTER A WITH BYTE AT ADDRESS TO DUMP
1850 CD 5D 18    CALL 185D     ;CALL CONVERT HEX TO ASCII CHARACTER AND PRINT ROUTINE
                                ;(HIGH)
1853 1A         LD A,(DE)     ;LOAD REGISTER A WITH BYTE AT ADDRESS TO DUMP (AGAIN)
1854 CD 61 18    CALL 1861     ;CALL CONVERT HEX TO ASCII CHARACTER AND PRINT ROUTINE
                                ;(LOW)
1857 13         INC DE        ;INCREMENT DE TO MOVE TO NEXT ADDRESS
1858 10 F1      DJNZ 184B     ;DECREASE REGISTER B AND REPEAT BYTE DUMP UNTIL B = 0
185A C3 31 18    JP 1831      ;JUMP TO CURRENT ADDRESS OUTPUT ROUTINE AND DO IT ALL
                                ;AGAIN...

```

CONVERT HEX DIGIT TO ASCII CHARACTER. ENTER AT 185D FOR HIGH NIBBLE CONVERT, ENTER AT 1861 FOR LOW NIBBLE CONVERT. ONCE HEX DIGIT IS SELECTED, IT IS USED AS AN INDEX INTO THE ASCII CHARACTER LOOKUP TABLE AT 186C

```

185D 1F         RRA           ;ROTATE RIGHT 4 TIMES TO MOVE
185E 1F         RRA           ;THE LOW NIBBLE TO THE HIGH
185F 1F         RRA           ;NIBBLE AND VICE VERSA
1860 1F         RRA           ;IE: '04' BECOMES '40'
1861 21 6C 18    LD HL,186C   ;LOAD HL WITH ASCII CHARACTER LOOKUP TABLE ADDRESS
1864 E6 0F      AND 0F        ;MASK OUT HIGH NIBBLE
1866 85         ADD A,L       ;ADD CURRENT L TO A
1867 6F         LD L,A        ;TO GET CORRECT INDEX
1868 7E         LD A,(HL)     ;LOAD REGISTER A WITH AN ASCII CHARACTER
1869 D3 06      OUT (06),A    ;OUTPUT ASCII CHARACTER TO PRINTER
186B C9         RET          ;EXIT

```

ASCII CHARACTER LOOKUP TABLE

```

186C 30 31 32 33 34 35 36 37      ;ASCII 0-7
1874 38 39 41 42 43 44 45 46      ;ASCII 8-F

```

BLANK FILL FOR 04 BYTES

```

187C FF FF FF FF      ;FILL

```

AUTO PRINT FROM ADDRESS ROUTINE. STARTING AT ADDRESS 0800, DATA IS SENT TO THE PRINTER UNTIL 'FF' IS FOUND. THE PRINTER MODE IS RESET TO TEXT MODE AND THEN A SOFT RESET IS PERFORMED.

```

1880 21 00 08      LD HL,0800 ;LOAD ADDRESS 0900 INTO HL

```

```

1883 7E          LD A,(HL)          ;LOAD REGISTER A WITH THE CONTENTS OF HL
1884 FE FF       CP FF              ;IS IT AN 'FF'
1886 20 05       JR NZ,188D         ;NO THEN JUMP TO 188D
1888 3E 11       LD A,11            ;LOAD A WITH TEXT MODE CONTROL CODE
188A D3 06       OUT (06),A         ;SEND IT TO THE PRINTER
188C C7          RST 00              ;RESTART THE TEC
188D D3 06       OUT (06),A         ;SEND ASCII OR CONTROL CODE TO PRINTER
188F 23          INC HL              ;MOVE TO NEXT ADDRESS
1890 18 F1       JR 1883            ;REPEAT FROM LINE 1883

```

BLANK FILL FOR 0E BYTES

```

1892 FF FF FF FF FF FF FF FF      ;FILL
199A FF FF FF FF FF FF FF

```

KEN'S PROGRAM. THIS IS SAMPLE PRINTER OUTPUT. IT IS IN TWO PARTS. THE FIRST PART DOES A FEW LINE FEEDS AND SETS UP THE COLOUR. THE SECOND PART PRINTS IN GRAPHICS MODE THE LETTERS "TEC-1" AND IT IS RUN 10 TIMES WITH A SLIGHTLY LOWER STARTING POINT. THIS GIVES IT A 3-D EFFECT. FIRST RUN IS PRINTED IN BLACK, NEXT 9 ARE IN RED. (NICE ONE KEN!)

PART 1 - LINE FEED 6 TIMES AND CHANGE PEN COLOUR TO BLACK

```

18A0 21 C3 18    LD HL,18C3        ;LOAD HL WITH START ADDRESS OF PRINTER CODES
18A3 7E          LD A,(HL)          ;LOAD PRINTER CODE INTO REGISTER A
18A4 FE FF       CP FF              ;IS IT 'FF' OR THE END OF THE CODES?
18A6 28 05       JR Z,18AD          ;YES, THEN MOVE TO PART 2
18A8 D3 06       OUT (06),A         ;SEND ASCII OR CONTROL CODE TO PRINTER
18AA 23          INC HL              ;MOVE TO NEXT ADDRESS
18AB 18 F6       JR 18A3            ;REPEAT FROM LINE 18A3

```

PART 2 - GRAPHICALLY PRINT THE LETTERS "TEC-1" 10 TIMES

```

18AD 06 0A       LD B,0A            ;LOAD REGISTER B WITH 0A. TO DO THE FOLLOWING 10 TIMES
18AF 21 CF 18    LD HL,18CF        ;LOAD HL WITH START ADDRESS OF PRINTER CODES
18B2 7E          LD A,(HL)          ;LOAD PRINTER CODE INTO REGISTER A
18B3 FE FF       CP FF              ;IS IT 'FF' OR THE END OF THE CODES?
18B5 28 05       JR Z,18BC          ;YES, THEN MOVE 18B2 AND REPEAT THE SEQUENCE
18B7 D3 06       OUT (06),A         ;SEND ASCII OR CONTROL CODE TO PRINTER
18B9 23          INC HL              ;MOVE TO NEXT ADDRESS
18BA 18 F6       JR 18B2            ;REPEAT FROM LINE 18B2
18BC 10 F1       DJNZ 18AF          ;REDUCE B, JUMP BACK TO 18AF TO REPEAT SEQUENCE IF !0
18BE 3E 11       LD A,11            ;LOAD A WITH TEXT MODE CONTROL CODE
18C0 D3 06       OUT (06),A         ;SEND TEXT MODE CODE TO PRINTER
18C2 C7          RST 00              ;RESTART THE TEC

```

KEY: CR = CARRIAGE RETURN, LF = LINE FEED, GFX = GRAPHICS MODE, EOS = END OF SEQUENCE
D = DRAW, M = MOVE, I = INITIALISE, C = SET COLOUR

PRINTER ASCII AND CONTROL CODES FOR PART 1

```

18C3 0D          ;CR
18C4 0A          ;LF
18C5 0A          ;LF
18C6 0A          ;LF
18C7 0A          ;LF
18C8 0A          ;LF
18C9 0A          ;LF
18CA 12          ;GFX
18CB 43          ;C          SET COLOUR
18CC 30          ;0          TO BLACK

```

| | | |
|---------|------|-------------------|
| 18CD 0D | ;CR | END OF COLOUR SET |
| 18CE FF | ;EOS | |

PRINTER ASCII AND CONTROL CODES FOR PART 2

| | | |
|------------|-----|---|
| 18CF 49 2C | ;I, | PRESENT PEN LOCATION IS TAKEN AS STARTING POINT |
|------------|-----|---|

| | | |
|---------------------|--------|-------------------------------|
| 18C1 44 | ;D | START OF DRAW |
| 18D2 33 32 30 2C 30 | ;320,0 | DRAW A LINE FROM 0,0 TO 320,0 |
| 18D7 0D | ;CR | END OF DRAW |

| | | |
|---------------------|--------|---------------|
| 18D8 4D | ;M | START OF MOVE |
| 18D9 31 32 30 2C 30 | ;120,0 | MOVE TO 120,0 |
| 18DE 0D | ;CR | END OF MOVE |

| | | |
|---------------------------|----------|-----------------------------------|
| 18DF 44 | ;D | START OF DRAW |
| 18E0 38 30 2C 2D 31 36 30 | ;80,-160 | DRAW A LINE FROM 120,0 TO 80,-160 |
| 18E7 0D | ;CR | END OF DRAW |

| | | |
|---------------------------|----------|-----------------|
| 18E8 4D | ;M | START OF MOVE |
| 18E9 32 32 30 2C 2D 38 30 | ;220,-80 | MOVE TO 220,-80 |
| 18F0 0D | ;CR | END OF MOVE |

| | | |
|------------------------------|-----------|---------------------------------------|
| 18F1 44 | ;D | START OF DRAW |
| 18F2 31 36 30 2C 2D 38 30 | ;160,-80 | DRAW A LINE FROM 220,-80 TO 160,-80 |
| 18F9 2C | ;; | |
| 18FA 31 34 30 2C 2D 31 26 30 | ;140,-160 | DRAW A LINE FROM 160,-80 TO 140,-160 |
| 1902 2C | ;; | |
| 1903 32 30 30 2C 2D 31 36 30 | ;200,-160 | DRAW A LINE FROM 140,-160 TO 200,-160 |
| 190B 0D | ;CR | END OF DRAW |

| | | |
|------------------------------|-----------|------------------|
| 190C 4D | ;M | START OF MOVE |
| 190D 31 35 30 2C 2D 31 32 30 | ;150,-120 | MOVE TO 150,-120 |
| 1915 0D | ;CR | END OF MOVE |

| | | |
|------------------------------|-----------|---------------------------------------|
| 1916 44 | ;D | START OF DRAW |
| 1917 32 30 30 2C 2D 21 32 30 | ;200,-120 | DRAW A LINE FROM 150,-120 TO 200,-120 |
| 191F 0D | ;CR | END OF DRAW |

| | | |
|---------------------------|----------|-----------------|
| 1920 4D | ;M | START OF MOVE |
| 1921 33 32 30 2C 2D 38 30 | ;320,-80 | MOVE TO 320,-80 |
| 1928 0D | ;CR | END OF MOVE |

| | | |
|------------------------------|-----------|---------------------------------------|
| 1929 44 | ;D | START OF DRAW |
| 192A 32 36 30 2C 2D 38 30 | ;260,-80 | DRAW A LINE FROM 320,-80 TO 260,-80 |
| 1931 2C | ;; | |
| 1932 32 34 30 2C 2D 31 36 30 | ;240,-160 | DRAW A LINE FROM 260,-80 TO 240,-160 |
| 193A 2C | ;; | |
| 193B 33 30 30 2C 2D 31 26 30 | ;300,-160 | DRAW A LINE FROM 240,-160 TO 300,-160 |
| 1943 0D | ;CR | END OF DRAW |

| | | |
|------------------------------|-----------|------------------|
| 1944 4D | ;M | START OF MOVE |
| 1945 33 36 30 2C 2D 31 32 30 | ;360,-120 | MOVE TO 360,-120 |
| 194D 0D | ;CR | END OF MOVE |

| | | |
|------------------------------|-----------|---------------------------------------|
| 194E 44 | ;D | START OF DRAW |
| 194F 34 30 30 2C 2D 31 32 30 | ;400,-120 | DRAW A LINE FROM 360,-120 TO 400,-120 |
| 1957 0D | ;CR | END OF DRAW |

| | | |
|---------|----|---------------|
| 1958 4D | ;M | START OF MOVE |
|---------|----|---------------|

```

1959 34 36 30 2C 2D 38 30      ;460,-80  MOVE TO 460,-80
1960 0D                        ;CR      END OF MOVE

1961 44                        ;D      START OF DRAW
1962 34 34 30 2C 2D 31 36 30    ;440,-160 DRAW A LINE FROM 460,-80 TO 440,-160
196A 0D                        ;CR      END OF DRAW

196B 4D                        ;M      START OF MOVE
196C 32 2C 2D 32                ;2,-2    MOVE TO 2,-2
1970 0D                        ;CR      END OF MOVE

1971 43                        ;C      SET COLOUR
1972 33                        ;3      TO RED
1973 0D                        ;CR      END OF COLOUR SET

1974 FF                        ;EOS

```

BLANK FILL FOR 0B BYTES

```

1975 FF FF FF FF FF FF FF FF    ;FILL
197D FF FF FF

```

MANUAL PRINT ROUTINE. ASCII OR CONTROL CODES ARE DIRECTLY TYPED
IN ONE BY ONE AND SENT TO THE PRINTER DIRECTLY. ONCE KEY IS PRESSED, IT
IS SAVED, THEN WAITS FOR ANOTHER KEY. THEN THESE TWO KEYS ARE COMBINED
TO CREATE ONE BYTE. THIS BYTE IS SENT TO THE PRINTER. TO EXIT, TEC
IS TO BE RESET.

```

1980 76      HALT      ;HALT THE CPU AND WAIT FOR KEYBOARD INPUT (NMI)
1981 ED 57    LD A,I    ;LOAD KEY INPUT INTO REGISTER A
1983 E6 0F    AND 0F    ;MASK OUT HIGH NIBBLE
1985 17      RLA      ;ROTATE LEFT 4 TIMES TO MOVE
1986 17      RLA      ;THE LOW NIBBLE TO THE HIGH
1987 17      RLA      ;NIBBLE AND VICE VERSA
1988 17      RLA      ;IE: '04' BECOMES '40'
1989 57      LD D,A    ;SAVE KEY IN D
198A 76      HALT      ;HALT THE CPU AND WAIT FOR KEYBOARD INPUT (NMI)
198B ED 57    LD A,I    ;LOAD KEY INPUT INTO REGISTER A
198D E6 0F    AND 0F    ;MASK OUT HIGH NIBBLE
198F 82      ADD A,D    ;ADD TO FIRST KEY
1990 D3 06    OUT (06),A ;SEND ASCII OR CONTROL CODE TO PRINTER
1992 18 EC    JR 1980   ;REPEAT MANUAL TYPING ROUTINE AGAIN

```

BLANK FILL FOR 0C BYTES

```

1994 FF FF FF FF FF FF FF FF    ;FILL
199C FF FF FF FF

```

MON2 PRINT-2 UPDATE

IF USING MON2, TWO ROUTINES CAN BE CALLED:

| PROGRAM | ADDR | DESCRIPTION |
|-----------------|--------|--|
| ----- | ---- | ----- |
| 1. MANUAL PRINT | - 19A0 | - TYPE IN ASCII OR CONTROL CODES DIRECTLY TO THE PRINTER |
| 2. HEX DUMP | - 19C0 | - GIVEN A START ADDRESS, OUTPUT THE DATA TO THE PRINTER. THIS ROUTINE WILL OUTPUT THE ADDRESS AND 8 BYTES PER LINE. |

MANUAL PRINT ROUTINE. ASCII OR CONTROL CODES ARE DIRECTLY TYPED IN ONE BY ONE AND SENT TO THE PRINTER DIRECTLY. ONCE KEY IS PRESSED, IT IS SAVED, THEN WAITS FOR ANOTHER KEY. THEN THESE TWO KEYS ARE COMBINED TO CREATE ONE BYTE. THIS BYTE IS SENT TO THE PRINTER. TO EXIT, TEC IS TO BE RESET.

```

19A0 76          HALT                ;HALT THE CPU AND WAIT FOR KEYBOARD INPUT (NMI)
19A1 3A E0 08    LD A,(08E0)          ;LOAD REGISTER A WITH THE KEY VALUE
19A4 E6 0F        AND 0F              ;MASK OUT HIGH NIBBLE
19A6 17          RLA                 ;ROTATE LEFT 4 TIMES TO MOVE
19A7 17          RLA                 ;THE LOW NIBBLE TO THE HIGH
19A8 17          RLA                 ;NIBBLE AND VICE VERSA
19A9 17          RLA                 ;IE: '04' BECOMES '40'
19AA 57          LD D,A              ;SAVE KEY IN D
19AB 76          HALT                ;HALT THE CPU AND WAIT FOR KEYBOARD INPUT (NMI)
19AC 3A E0 08    LD A,(08E0)          ;LOAD REGISTER A WITH THE KEY VALUE
19AF E6 0F        AND 0F              ;MASK OUT HIGH NIBBLE
19B1 82          ADD A,D              ;ADD TO FIRST KEY
19B2 D3 06        OUT (06),A          ;SEND ASCII OR CONTROL CODE TO PRINTER
19B4 18 EA        JR 19A0             ;REPEAT MANUAL TYPING ROUTINE AGAIN

```

BLANK FILL FOR 0A BYTES

```

19B6 FF FF FF FF FF FF FF FF FF FF ;FILL

```

START OF HEX DUMP ROUTINE.

```

19C0 3E 0D        LD A,0D            ;LOAD CARRIAGE RETURN CODE TO REGISTER A
19C2 D3 06        OUT (06),A          ;OUTPUT CR TO PRINTER
19C4 3E 0A        LD A,0A            ;LOAD LINE FEED CODE TO REGISTER A
19C6 D3 06        OUT (06),A          ;OUTPUT LF TO PRINTER
19C8 3E 29        LD A,29            ;LOAD REGISTER A WITH LED '0'
19CA 21 D8 08     LD HL,08D8          ;LOAD HL WITH LED SCREEN BUFFER START ADDRESS
19CD 06 06        LD B,06            ;SIX LED'S TO OUTPUT TO
19CF 77          LD (HL),A           ;SAVE A '0' TO SCREEN BUFFER
19D0 23          INC HL              ;MOVE TO NEXT BUFFER
19D1 10 FC        DJNZ 19CF           ;RETURN UNTIL ALL SIX BUFFERS ARE FILLED
19D3 CD 00 1A     CALL 1A00           ;KEYBOARD READ AND DISPLAY ROUTINE. RETURNS KEY IN A
19D6 32 D8 08     LD (08D8),A         ;SAVE KEY READ INTO FIRST ADDRESS BUFFER
19D9 CD 00 1A     CALL 1A00           ;KEYBOARD READ AND DISPLAY ROUTINE
19DC 32 D9 08     LD (08D9),A         ;SAVE KEY READ INTO SECOND ADDRESS BUFFER
19DF CD 00 1A     CALL 1A00           ;KEYBOARD READ AND DISPLAY ROUTINE
19E2 32 DA 08     LD (08DA),A         ;SAVE KEY READ INTO FIRST THIRD BUFFER
19E5 CD 00 1A     CALL 1A00           ;KEYBOARD READ AND DISPLAY ROUTINE
19E8 32 DB 08     LD (08DB),A         ;SAVE KEY READ INTO FOURTH ADDRESS BUFFER
19EB CD D8 01     CALL 01D8           ;CALL MULTIPLEX DISPLAY ROUTINE
19EE CD 89 02     CALL 0289           ;CALL GET EDITOR ADDRESS ROUTINE. RETURNS ADDR IN BC
19F1 50          LD D,B              ;LOAD DE WITH BC
19F2 59          LD E,C              ;DE IS USED IN THE PRINT-1 HEX DUMP ROUTINE

```

19F3 C3 31 18 JP 1831 ;JUMP TO PRINT-1 HEX DUMP ROUTINE AND CONTINUE FOREVER

BLANK FILL FOR 0A BYTES

19F6 FF FF FF FF FF FF FF FF FF FF ;FILL

KEYBOARD READ ROUTINE. LOOPS UNTIL A KEY IS PRESSED. WHEN A KEY IS PRESSED
NMI IS TRIGGERED AND THE KEY IS STORED IN LOCATION 08E0. IF A KEY IS PRESSED
IT IS DISPLAYED ON THE LED'S, A TONE IS PLAYED AND KEY PRESSED IS PLACED IN REGISTER A

| | | |
|---------------|-------------|---|
| 1A00 3E FF | LD A,FF | ;LOAD 'FF' IN REGISTER A. 'FF' IS THE DEFAULT VALUE |
| 1A02 32 E0 08 | LD (08E0),A | ;STORE 'FF' IN KEY ADDRESS LOCATION 08E0 |
| 1A05 CD A0 02 | CALL 02A0 | ;UPDATE LED DISPLAY |
| 1A08 3A E0 08 | LD A,(08E0) | ;LOAD KEY ADDRESS LOCATION TO REGISTER A |
| 1A0B FE FF | CP FF | ;COMPARE TO 'FF' |
| 1A0D 28 F6 | JR Z,1A05 | ;REPEAT UNTIL KEY IS PRESSED |
| 1A0F E6 0F | AND 0F | ;MASK OUT HIGH NIBBLE |
| 1A11 C6 01 | ADD A,01 | ;ADD ONE. THIS IS BECAUSE THE TONE ROUTINE |
| 1A13 CD 70 01 | CALL 0170 | ;EXITS ON ZERO. CALL TONE ROUTINE |
| 1A16 D6 01 | SUB 01 | ;RESTORE KEY PRESSED VALUE |
| 1A18 C9 | RET | ;EXIT |

BLANK FILL FOR 07 BYTES

1A19 FF FF FF FF FF FF FF ;FILL

JMON PRINT-3 UPDATE

IF USING JMON, THREE ROUTINES CAN BE CALLED:

| PROGRAM | ADDR | DESCRIPTION |
|-----------------|------|--|
| ----- | ---- | ----- |
| 1. HEX DUMP | 1A20 | GIVEN A START AND END ADDRESS, OUTPUT THE DATA TO THE PRINTER. THIS ROUTINE WILL OUTPUT THE ADDRESS AND 8 BYTES PER LINE. |
| 2. MANUAL PRINT | 1AA0 | TYPE IN ASCII OR CONTROL CODES DIRECTLY TO THE PRINTER |
| 3. AUTO PRINT | 1AC0 | OUTPUT BYTES FROM ADDRESS 0900 UNTIL BYTE 'FF' IS REACHED |

ENTRY POINT TO THE HEX DUMP. FIRST IS THE PERIMETER HANDLER SETUP ROUTINE.

| | | |
|---------------|--------------|--|
| 1A20 21 34 1A | LD HL,1A34 | ;SOURCE DATA FOR JMON PERIMETER HANDLER |
| 1A23 11 80 08 | LD DE,0880 | ;DESTINATION OF COMMAND STRING |
| 1A26 01 0A 00 | LD BC,000A | ;COMMAND STRING LENGTH 10 BITES |
| 1A29 ED B0 | LDIR | ;COPY SOURCE TO DESTINATION |
| 1A2B 21 00 00 | LD HL,0000 | ;LOAD HL WITH 0000 |
| 1A2E 22 9C 08 | LD (089C),HL | ;CLEAR OUT THIRD ADDRESS POINT? (ONLY 2 USED!) |
| 1A31 C3 44 00 | JP 0044 | ;CALL THE JMON PERIMETER HANDLER ROUTINE |
| | | ;USER THEN ENTERS START AND END ADDRESS |
| | | ;THEN PRESSES GO WHICH JUMPS TO 1A50 |

PERIMETER COMMAND STRING FOR START AND END ADDRESS. BYTES 1&2 ARE SIGNATURE BYTES (OPTIONAL), BYTES 3&4 IS ADDRESS OF THE DISPLAY CODES 1A3E, BYTES 5&6 IS THE INPUT WINDOW POINTER SET AT 0899 (HIGH ORDER BYTE), BYTE 7 IS THE FIRST DISPLAY CODE, BYTE 8 IS THE SIZE OF THE DISPLAY CODES, BYTES 9&10 IS THE JUMP ADDRESS WHEN GO IS PRESSED. JUMPS TO 1A50

1A34 00 00 3E 1A 99 08 00 01 50 1A

PERIMETER HANDLER DATA DISPLAYS

| | | |
|------------|------|----------------|
| 1A3E 04 A7 | "-S" | ;START ADDRESS |
| 1A40 04 C7 | "-E" | ;END ADDRESS |
| 1A42 04 EB | "-O" | ;NOT USED |

BLANK FILL FOR 0C BYTES

| | |
|------------------------------|-------|
| 1A44 FF FF FF FF FF FF FF FF | ;FILL |
| 1A4C FF FF FF FF | |

FROM THE START ADDRESS (STORED IN 0898) TO THE END ADDRESS (STORED IN 089A), OUTPUT THE ADDRESS AND 8 BYTES OF DATA PER LINE. EG:

-> 1A50 3E 0D D3 06 2A 98 08 7C
-> 1A58 CD 82 1A 7D CD 82 1A 06
-> 1A60 08 C5 3E 20 D3 06 7E CD
-> 1A68 82 1A 23 C1 10 F3 3E 0D

| | | |
|---------------|--------------|---|
| 1A50 3E 0D | LD A,0D | ;LOAD CARRIAGE RETURN CODE TO REGISTER A |
| 1A52 D3 06 | OUT (06),A | ;OUTPUT CR TO PRINTER |
| | | ;OUTPUT THE CURRENT ADDRESS TO THE PRINTER |
| 1A54 2A 98 08 | LD HL,(0898) | ;LOAD HL WITH START ADDRESS SET BY PERIMETER CODE |
| 1A57 7C | LD A,H | ;LOAD REGISTER A WITH H |
| 1A58 CD 82 1A | CALL 1A82 | ;CALL CONVERT REGISTER A TO ASCII AND SEND TO PRINTER |
| 1A5B 7D | LD A,L | ;LOAD REGISTER A WITH L |

| | | | |
|------|-------------|--------------|---|
| 1A5C | CD 82 1A | CALL 1A82 | ;CALL CONVERT REGISTER A TO ASCII AND SEND TO PRINTER |
| 1A5F | 06 08 | LD B,08 | ;LOAD B WITH 8 DATA BYTES TO PRINT |
| 1A61 | C5 | PUSH BC | ;SAVE BC |
| 1A62 | 3E 20 | LD A,20 | ;LOAD REGISTER A WITH AN ASCII SPACE " " |
| 1A64 | D3 06 | OUT (06),A | ;OUTPUT SPACE TO PRINTER |
| 1A66 | 7E | LD A,(HL) | ;LOAD BYTE AT CURRENT ADDRESS TO REGISTER A |
| 1A67 | CD 82 1A | CALL 1A82 | ;CALL CONVERT REGISTER A TO ASCII AND SEND TO PRINTER |
| 1A6A | 23 | INC HL | ;MOVE TO NEXT ADDRESS |
| 1A6B | C1 | POP BC | ;RESTORE BC |
| 1A6C | 10 F3 | DJNZ 1A61 | ;REDUCE B AND RETURN TO 1A61 IF NOT ZERO |
| 1A6E | 3E 0D | LD A,0D | ;LOAD CARRIAGE RETURN CODE TO REGISTER A |
| 1A70 | D3 06 | OUT (06),A | ;OUTPUT CR TO PRINTER |
| 1A72 | 3E 0A | LD A,0A | ;LOAD LINE FEED CODE TO REGISTER A |
| 1A74 | D3 06 | OUT (06),A | ;OUTPUT LF TO PRINTER |
| 1A76 | ED 5B 9A 08 | LD DE,(089A) | ;LOAD DE WITH END ADDRESS |
| 1A7A | E5 | PUSH HL | ;SAVE CURRENT ADDRESS |
| 1A7B | B7 | OR A | ;RESET CARRY FLAG |
| 1A7C | ED 52 | SBC HL,DE | ;SUBTRACT DE AND THE CARRY FLAG FROM HL |
| 1A7E | E1 | POP HL | ;RESTORE HL |
| 1A7F | 38 D6 | JR C,1A57 | ;IF CARRY IS SET (DE > HL) THEN REPEAT THE |
| | | | ;PRINT PROCESS |
| 1A81 | C9 | RET | ;RETURN AND EXIT |

SMART ROUTINE TO CONVERT A BYTE TO ASCII

| | | | |
|------|----------|------------|---|
| 1A82 | F5 | PUSH AF | ;SAVE AF FOR LATER USE |
| 1A83 | 0F | RRCA | ;ROTATE RIGHT 4 TIMES TO MOVE |
| 1A84 | 0F | RRCA | ;THE LOW NIBBLE TO THE HIGH |
| 1A85 | 0F | RRCA | ;NIBBLE AND VICE VERSA |
| 1A86 | 0F | RRCA | ;IE: '4C' BECOMES 'C4' |
| 1A87 | CD 8B 1A | CALL 1A8B | ;CALL CONVERT LOW NIBBLE TO ASCII AND PRINT ROUTINE |
| 1A8A | F1 | POP AF | ;RESTORE AF |
| 1A8B | E6 0F | AND 0F | ;MASK HIGH NIBBLE TO KEEP ONLY THE LOW NIBBLE |
| 1A8D | C6 90 | ADD A,90 | ;ADD 0X90 TO A. CARRY WILL BE SET IF HEX IS A-F |
| 1A8F | 27 | DAA | ;CONVERT TO DECIMAL |
| 1A90 | CE 40 | ADC A,40 | ;ADD 40 WITH CARRY TO A |
| 1A92 | 27 | DAA | ;CONVERT TO DECIMAL |
| 1A93 | D3 06 | OUT (06),A | ;OUTPUT ASCII TO PRINTER |
| 1A95 | C9 | RET | ;EXIT |

BLANK FILL FOR 0A BYTES

1A96 FF FF FF FF FF FF FF FF FF FF ;FILL

MANUAL PRINT ROUTINE. ASCII OR CONTROL CODES ARE DIRECTLY TYPED IN ONE BY ONE AND SENT TO THE PRINTER DIRECTLY. ONCE KEY IS PRESSED, IT IS SAVED, THEN WAITS FOR ANOTHER KEY. THEN THESE TWO KEYS ARE COMBINED TO CREATE ONE BYTE. THIS BYTE IS SENT TO THE PRINTER. TO EXIT, TEC IS TO BE RESET.

| | | | |
|------|-------|------------|--|
| 1AA0 | CF | RST 08 | ;CALL JMON KEY WAIT AND HANDLE ROUTINE |
| 1AA1 | E6 0F | AND 0F | ;MASK OUT HIGH NIBBLE |
| 1AA3 | 07 | RLCA | ;ROTATE FOUR BITS TO THE LEFT |
| 1AA4 | 07 | RLCA | ;THIS PLACES THE CONTENTS IN THE HIGH NIBBLE |
| 1AA5 | 07 | RLCA | ;IE: 02 WILL BECOME 20 |
| 1AA6 | 07 | RLCA | ; |
| 1AA7 | 57 | LD D,A | ;SAVE FIRST KEY PRESS IN REGISTER D |
| 1AA8 | CF | RST 08 | ;CALL JMON KEY WAIT AND HANDLE ROUTINE |
| 1AA9 | E6 0F | AND 0F | ;MASK OUT HIGH NIBBLE |
| 1AAB | 82 | ADD A,D | ;ADD TO FIRST KEY |
| 1AAC | D3 06 | OUT (06),A | ;SEND ASCII OR CONTROL CODE TO PRINTER |

1AAE 18 F0 JR 1AA0 ;REPEAT MANUAL TYPING ROUTINE AGAIN.

CODE HERE IS NOT REACHED!

1AB0 D3 06 OUT (06),A
1AB2 18 EC JR 1AA0

BLANK FILL FOR 0C BYTES

1AB4 FF FF FF FF FF FF FF FF ;FILL
1ABC FF FF FF FF

AUTO PRINT FROM ADDRESS ROUTINE. STARTING AT ADDRESS 0900, DATA IS SENT TO THE PRINTER
UNTIL 'FF' IS FOUND. THE PRINTER MODE IS RESET TO TEXT MODE AND THEN A SOFT RESET
IS PERFORMED.

1AC0 21 00 09 LD HL,0900 ;LOAD ADDRESS 0900 INTO HL
1AC3 7E LD A,(HL) ;LOAD REGISTER A WITH THE CONTENTS OF HL
1AC4 FE FF CP FF ;IS IT AN 'FF'
1AC6 20 05 JR NZ,1ACD ;NO THEN JUMP TO 1ACD
1AC8 3E 11 LD A,11 ;LOAD A WITH TEXT MODE CONTROL CODE
1ACA D3 06 OUT (06),A ;SEND IT TO THE PRINTER
1ACC C7 RST 00 ;RESTART THE TEC
1ACD D3 06 OUT (06),A ;SEND ASCII OR CONTROL CODE TO PRINTER
1ACF 23 INC HL ;MOVE TO NEXT ADDRESS
1AD0 18 F1 JR 1AC3 ;REPEAT FROM LINE 1AC3

BLANK FILL FOR 1E BYTES

1AD2 FF FF FF FF FF FF FF FF ;FILL
1ADA FF FF FF FF FF FF FF FF
1AE2 FF FF FF FF FF FF FF FF
1AEA FF FF FF FF FF FF